Predicate Logic Proofs in PVS

© 2001 M. Lawford

Outline

• Review: Order of precedence & Dealing with quantifiers

• Universal closure of sentence forms in PVS

• (SKOLEM!) and (INST . . . ) - PVS commands for eliminating quantifiers

• Example: Putting it all together

• Rules for dealing with =: (EXPAND . . . ) & (REPLACE . . . )

Order of Precedence & Parenthesis

Recall: We use precedence of logical operators and associativity of ∧, ∨, ↔ to drop parentheses. It is understood that this is shorthand for the fully parenthesized expressions.

Huth+Ryan uses order of precedence: ∀ ∃ ¬ , ∧ ∨ , → ↔ PVS uses order of precedence: ¬, ∧, ∨, →, ↔, ∀ ∃ ∀xP (x) → ∃yQ(x, y) ∧ P (y) becomes: In Huth+Ryan: (∀xP (x)) → ((∃yQ(x, y)) ∧ P (y)) In PVS: ∀x(P (x) → (∃y(Q(x, y) ∧ P (y))))

Review

As we will see, the PVS commands that deal with quantifiers and equality can all be understood in terms of the rules we already know.

introduction elmination

∀ x0 . ϕ[x0/x] —————– ∀i ∀xϕ ∀xϕ ϕ[t/x] ∀e

∃ ϕ[t/x] ∃xϕ ∃i ∃xϕ x0 ϕ[x0/x] . χ ——————— ∃e χ

= t = t = i t1 = t2 ϕ[t1/x] —————– = e ϕ[t2/x]

Removing & Adding ∀x

Rule ∀e: If � ⊢ ∀xϕ and ϕ[t/x] is a valid substitution then � ⊢ ϕ[t/x]. Rule ∀i: If � ⊢ ϕ[x0/x], and 1. x0 ∉ F V (�), 2. and ϕ[x0/x] is a valid substitution. Then � ⊢ ∀xϕ.

Adding and Removing ∃x

Rule ∃i: If � ⊢ ϕ[t/x] and ϕ[t/x] is a valid substitution then � ⊢ ∃xϕ. Rule ∃e: If �, ϕ[x0/x] ⊢ χ, and 1. ϕ[x0/x] is a valid substitution, and 2. x0 ∉ F V (�) ∪ F V (χ). Then �, ∃xϕ ⊢ χ. 4

PVS Declarations

When proving things in propositional logic in Huth+Ryan, our universe A is a nonempty set or “type” of object.

basics : THEORY
  BEGIN

 A:TYPE+ % Nonempty universe
 x, y:VAR A % x and y are variables of type A
 a, b: A % a, b are constant elements of A
 P: PRED[A] % P is a unary predicate
 Q: PRED[[A,A]] % Q is a binary predicate
 f: [A -> A] % f is a function of 1 arg
 h(x):A % h is also a function of 1 arg
 g: [[A,A]-> A] % 2-ary function

  END basics

Note: PRED[A] is equivalent to:

PRED: TYPE = [A -> bool]

5

Universal Closure in PVS

Def: For ϕ with free variables x1, . . . , xn, the formula ∀x1∀x2 . . . ∀xnϕ is the universal closure of ϕ. Note that M |= ϕ iff M |= ∀x1∀x2 . . . ∀xnϕ (follows immediately definition of |=.) PVS uses this as a short cut to implicitly quantify theorem statements. E.g.

x,y,z:VAR nat
f(x,y):nat = x + y
T1: THEOREM f(x,y)=f(y,x)

in prover becomes:

T1 :
  |-------
{1} (FORALL (x: nat, y: nat): f(x, y) = f(y, x))
Rule?

6

Universal Closure in PVS (cont.)

Note: You must be careful stating negation of formulas!

Consider the following PVS:

x:VAR nat
 P1: PROPOSITION x+x>x
 WrongNotP1: PROPOSITION NOT(x+x>x)
 NotP1: PROPOSITION NOT(FORALL (x:nat): (x+x>x))

Neither P1 nor WrongNotP1 is provable but NotP1 is provable. Why? Try proving WrongNotP1 and due to universal closure you get:

WrongNotP1 :
  |-------
{1} FORALL (x: nat): NOT (x + x > x)
Rule?

7

Predicate Logic Proofs in PVS

Predicate logic proofs are just propositional logic proofs with new rules for eliminating quantifiers.

Still use the commands for propositional rules: (FLATTEN), (SPLIT) & (BDDSIMP).

And add new PVS commands:

(SKOLEM!) does ∃e and ∀i.

(INST eq# “term”) does ∀e and ∃i.

PVS commands: (SKOLEM!)

Let x0 be a new “variable” (a.k.a skolem constant) not appearing in any of the formulas of sequents of the sequent.

(SKOLEM!) uses ∃e to eliminate ∃x in a premises: ϕ1 ϕ2 . ∃xϕ ψ1 ψ2 .

(SKOLEM!) =⇒ ϕ1 ϕ2 . ϕ[x0/x] ψ1 ψ2 . or ∀i to eliminate ∀x in a conclusion: ϕ1 ϕ2 . ψ1 ψ2 . ∀xϕ

(SKOLEM!) =⇒ ϕ1 ϕ2 . ψ1 ψ2 . ϕ[x0/x]

How (SKOLEM!) uses Rule ∀i

Rule ∀i: If � ⊢ ϕ[x0/x], x0 ∉ F V (�) and ϕ[x0/x] is valid, then � ⊢ ∀xϕ.

To try to prove � ⊢ ∀xϕ, (SKOLEM!) says PVS can try to prove � ⊢ ϕ[x0/x] for some “new” x0 so that i) x0 ∉ F V (�) ∪ F V (ϕ), and ii) ϕ[x0/x] is a valid substitution. Why? If � ⊢ ϕ[x0/x] then � ⊢ ∀xϕ, by Rule ∀i. Why make sure x0 ∉ F V (ϕ) too? E.g. Consider formula ∀xP (x, x0).

Why require ϕ[x1/x] is a valid substitution? E.g. Consider formula ∀x∃x0(x < x0)

PVS commands: (INST . . . )

Below let t be a term such that ϕ[t/x] is valid.

(INST . . . ) uses EG to “remove”∗ ∃x in a conclusion: −1 ϕ1 −2 ϕ2 . . 1 ψ1 2 ψ2 . . n ∃xϕ

(INST n “t”) =⇒ ϕ1 ϕ2 . ψ1 ψ2 . ϕ[t/x]

∗The original quantified formula is still available to use in proofs. Its just “hidden”. Use the PVS menu command “M-x show-hidden-formulas” to see hidden formulas and the prover command (REVEAL eq#) to use a hidden equation.

PVS commands: (INST . . . )

Below let t be a term such that ϕ[t/x] is valid.

(INST ) uses ∀e to “remove”∗ ∀x in a premises: −1 ϕ1 −2 ϕ2 . . −n ∀xϕ 1 ψ1 2 ψ2 . .

(INST -n “t”) =⇒ ϕ1 ϕ2 . ϕ[t/x] ψ1 ψ2 .

∗Again, the original quantified formula is still available to use in proofs. Its just “hidden”.

Use (INST-CP -n “t”) to keep a copy of original formula in sequent.

How (INST . . . ) uses Rule ∀e

Rule ∀e: If � ⊢ ∀xϕ and ϕ[t/x] is a valid substitution then � ⊢ ϕ[t/x].

To try to prove �, ∀xϕ ⊢ ψ, (INST -n “t”) says PVS can try to prove �, ϕ[t/x] ⊢ ψ for some t such that ϕ[t/x] is a valid substitution. Why? If �, ϕ[t/x] ⊢ ψ then �, ∀xϕ ⊢ ψ, because �, ∀xϕ ⊢ ϕ[t/x], by Rule ∀e So you can finish proof of ψ from �, ∀xϕ by using proof of ψ from �, ϕ[t/x].

Putting it all together

Ex 2 (revisited): Use PVS to prove inconsistency of: �′′ = {∀x(P (x)∨Q(x)), ∀y(¬P (y) → ¬Q(y)), ∃x¬P (x)}

predicate : THEORY
  BEGIN
  A:NONEMPTY_TYPE
  x,y: VAR A
  P,Q:PRED[A]
  I1: PROPOSITION (FORALL x:P(x) OR Q(x))&
    (FORALL y:NOT P(y) IMPLIES NOT Q(y)) &
    (EXISTS x:NOT P(x)) IMPLIES FALSE
  END predicate

I1 :

  |-------
{1} (FORALL x: P(x) OR Q(x))
          & (FORALL y: NOT P(y) IMPLIES NOT Q(y)) &
          (EXISTS x: NOT P(x)) IMPLIES FALSE

Rule? (FLATTEN)
Applying disjunctive simplification to flatten sequent,
this simplifies to:
I1 :

{-1} (FORALL x: P(x) OR Q(x))
{-2} (FORALL y: NOT P(y) IMPLIES NOT Q(y))
{-3} (EXISTS x: NOT P(x))
  |-------

Rule? (SKOLEM!)
Skolemizing,
this simplifies to:
I1 :

[-1] (FORALL x: P(x) OR Q(x))
[-2] (FORALL y: NOT P(y) IMPLIES NOT Q(y))
  |-------
{1} P(x!1)

.

Rule? (INST -1 "x!1")
Instantiating the top quantifier in -1 with the terms:
 x!1,
this simplifies to:
I1 :

{-1} P(x!1) OR Q(x!1)
[-2] (FORALL y: NOT P(y) IMPLIES NOT Q(y))
  |-------
[1] P(x!1)

Rule? (INST - "x!1")
Instantiating the top quantifier in - with the terms:
 x!1,
this simplifies to:
I1 :

[-1] P(x!1) OR Q(x!1)
{-2} NOT P(x!1) IMPLIES NOT Q(x!1)
  |-------
[1] P(x!1)

Rule? (BDDSIMP)
Applying bddsimp,
this simplifies to:
I1 :

{-1} FALSE
  |-------

which is trivially true.
Q.E.D.

16

PVS Commands for Dealing with =

(EXPAND "t1") and (EXPAND "t1" "t2" ...)

equality: THEORY
  BEGIN
  x,y:VAR real
  a:real=1
  f(x,y):real = x+y
  g(x,y):real = x+y

  Ia: THEOREM f(y,a)=g(y,1)
  END equality

To prove THEOREM Ia you can just use (SKOLEM!) to eliminate universal quantifiers and then use variants of the (EXPAND ...) command to expand definitions (EXPAND* "f" "g") (EXPAND "a").

PVS Commands for Dealing with =

Q: How do you use premises with top level “=” in PVS that are not definitions? A: The PVS equivalent of Huth+Ryan’s = e rule, a.k.a. “Substitution of Equals”, is:

(REPLACE -n *)

If equation -n in the premises is of the form tL = tR The command makes all valid substitutions of tR for tL in all other formulas of the sequent! Changing the above command to

(REPLACE -n * RL)

would replace right-to-left, performing all valid substitutions of tL for tR. 18

Example: Rubin p.244 E11

equal11 : THEORY
  BEGIN
  A:TYPE+
  P:PRED[A]
  A,B,C,D:PRED[A]
  x,y : VAR A
  E11: THEOREM (FORALL x,y:A(x)&B(y)=> x=y)
 &(EXISTS x:A(x)&C(x)) & (EXISTS x:B(x)&D(x))
    =>(EXISTS x:C(x)&D(x))
  END equal11

Using a combination of (BDDSIMP), (SKOLEM!) and (INST?) reduces E11 to sequent

{-1} A(x!1)
{-2} B(x!2)
{-3} x!1 = x!2
{-4} C(x!1)
{-5} D(x!2)
  |-------
{1} D(x!1)

19

Now you can finish off the proof by replacing

x!1 by x!2 as follows:

Rule? (REPLACE -3 * LR)
Replacing using formula -3,
this simplifies to:
E11 :

{-1} A(x!2)
[-2] B(x!2)
[-3] x!1 = x!2
{-4} C(x!2)
[-5] D(x!2)
  |-------
{1} D(x!2)

which is trivially true.
Q.E.D.

Failed Proofs & Counter Examples

Suppose you are asked if the a proof exists for the following sequent: ∃x[E(x) ∧ ∀y(F (y) → G(x, y))], ∀x∀y[E(x) → (G(x, y) ↔ H(y))] ? ⊢ ∀x(F (x) ↔ H(x)) Putting this into PVS and trying to prove it results can results the sequent:

[-1] E(x!1)
{-2} G(x!1, x!2)
{-3} H(x!2)
  |-------
{1} F(x!2)

Rule?

This sequent would be true if E(x1), G(x1, x2), H(x2) ⊢ F (x2) or ⊢ E(x1) ∧ G(x1, x2) ∧ H(x2) → F (x2)

Failed Proofs & Counter Examples

Just as we had for propositional logic, syntax and semantics agree so � ⊢ ψ iff � |= ψ Thus ⊢ E(x1) ∧ G(x1, x2) ∧ H(x2) → F (x2) iff |= E(x1) ∧ G(x1, x2) ∧ H(x2) → F (x2) which by definition of |= holds iff |= ∀x1∀x2(E(x1) ∧ G(x1, x2) ∧ H(x2) → F (x2)) but this would mean that for every model M M |= ∀x1∀x2(E(x1)∧G(x1, x2)∧H(x2) → F (x2)) But we can find an M such that: M |̸= ∀x1∀x2(E(x1)∧G(x1, x2)∧H(x2) → F (x2)) Check that this model provides a counter example to the original sequent! 22